
ABSTRACT  

Oncology studies are among the most complex and challenging therapeutic areas for statistical 
programmers. Navigating the efficacy analysis for these studies can be particularly daunting for those 
new to the field. This paper aims to provide a comprehensive overview of the most common efficacy 
outputs requested for early phase oncology trials, along with basic codes and techniques to assist in 
programming these outputs.  

The paper will discuss common efficacy outputs generated for solid tumor studies, which are the most 
prevalent type of oncology trials. Solid tumor studies typically follow RECIST (Response Evaluation 
Criteria in Solid Tumor) guidelines. This paper will focus on the roadmap to efficacy analysis, including 
the creation of efficacy datasets (ADTR, ADRS, ADTTE) and the parameters used to create outputs for 
duration of response, Kaplan-Meier estimates for survival analysis, waterfall plots, spider plots, and 
swimmer plots. 

Finally, the paper will provide relevant SAS® code for all these methods, equipping readers with the 
knowledge and tools needed to create tables and figures for oncology studies. 

INTRODUCTION  

Programming the efficacy analysis for early phase oncology trials is a journey. We start with a set 
of guidelines and raw data as collected on the CRFs, and we need to arrive at the TLFs shown at 
the end of this paper to obtain meaningful results in an understandable format. In this paper, we 
hope to provide a guide on this journey, highlighting best practices and illuminating potholes to 
avoid along the way. Our route, displayed in  

Figure 1. Route to Efficacy Analysis 

Figure 1, starts with the mapping of raw data to the relevant SDTM domains, and takes us all the way to 
the TLF outputs. The first part of this paper will focus on the details of the SDTM and how it is mapped 
from the CRFs. In the second part, we start with the TLFs and work backwards to show how we derive 
the key parameters in ADaM and program them to create the desired output. We hope you enjoy the 
journey!  

RULES OF THE ROAD: RECIST (RESPONSE EVALUATION CRITERIA IN SOLID 
TUMORS) 

RECIST provides a standardized framework for assessing tumor response in solid tumor oncology trials 
(Eisenhauer et. al 2008). It defines objective criteria for evaluating disease response. RECIST criteria 
include three components: 

• Target lesions are specific, measurable lesions selected for detailed monitoring during treatment. 
They must be measurable, typically with a longest diameter of at least 10 mm on CT scans (or 15 mm 
for lymph nodes). Only a limited number of lesions (up to 5 in total, with a maximum of 2 per organ) 
are chosen as target lesions. 

• Non-target lesions include all other lesions that do not meet the criteria for target lesions. 
Measurements are not required, and these lesions should be followed as ‘present’, ‘absent’, or in rare 
cases ‘unequivocal progression’. 

• New Lesions include all lesions not identified at baseline that develop during the study. The 
presence of a new lesion indicates disease progression.  The finding of a new lesion should be 
unequivocal. If equivocal, it should be followed for subsequent visits until it is confirmed that it is 
definitely a new lesion. 

The primary RECIST-defined response categories include: 

• Complete Response (CR): Disappearance of all target and non-target lesions with no new lesions 
appearing. 

• Partial Response (PR): At least a 30% decrease in the sum of diameters of target lesions compared 
to baseline with no progression of non-target lesions and no new lesions. 

• Stable Disease (SD): Neither sufficient shrinkage to qualify as PR nor sufficient increase to qualify as 
Progressive Disease. 

• Progressive Disease (PD): At least a 20% increase in the sum of diameters of target lesions taking 
as reference the smallest sum on study or unequivocal progression of existing non-target lesions or 
appearance of new lesions 

These response categories are essential for evaluating treatment efficacy and are captured in the RS 
dataset at each assessment visit. 

PAVING THE WAY: ONCOLOGY-SPECIFIC SDTM DOMAINS  

The TU (Tumor Identification), TR (Tumor Results) and RS (Response) domains are unique to oncology 
studies. This section will describe those domains in detail and show how the data is mapped from the 
CRFs to the SDTM domains.  

RELATIONSHIP BETWEEN RS, TR, AND TU DATASETS 

These datasets are closely related and provide lesion-level information essential for deriving tumor 
response assessments: 

• TU (Tumor Identification Dataset): Catalogs all tumors assessed in a study, specifying their 
anatomical locations and whether they are target or non-target lesions. 

• TR (Tumor Results Dataset): Contains quantitative measurements of individual lesions identified in 
TU, recording lesion size (e.g., longest diameter for target lesions) and changes over time to calculate 
tumor burden. 

• RS (Response Dataset): Aggregates lesion measurements from TR and applies RECIST criteria to 
determine overall tumor response. 

The sequential relationship among these datasets is as follows: 

1. TU defines the lesions tracked throughout the study. 

2. TR provides longitudinal measurements of these lesions, allowing for the calculation of percent 
changes from baseline. 

3. RS summarizes overall response at each time point by applying RECIST-defined response 
categories based on TR measurements. 

By transforming lesion-level data into patient-level summaries, these datasets are essential for evaluating 
the success of oncology treatments. 

TUMOR IDENTIFICATION (TU) 

The Tumor Identification (TU) domain is designed to uniquely identify tumors tracked during a study. 
Initial identification is typically done by an assessor at a baseline visit and is not repeated at every visit, 
resulting in one record per tumor per subject. Post-baseline records are included for new tumors, tumors 
that split, or merged tumors. Key variables in the TU domain include: 

• TULNKID: Links identified tumors to assessment results. 

• TUTEST and TUTESTCD: Represent the test for tumor identification, with "TUMIDENT" assigned for 
tumor identification. 

• TUGRPID: Links related records for split or merged tumors. 

• TULOC: Specifies the anatomical location of the tumor. 

• TUMETHOD: Indicates the assessment method (e.g., MRI, CT). 

• TUORRES: Stores the classification result of the tumor (e.g., TARGET, NONTARGET, NEW). 

• TUEVAL: Collects the role of the evaluator (e.g., INVESTIGATOR, INDEPENDENT ASSESSOR). 

These variables ensure accurate tracking and classification of tumors throughout the study. Tumor details 
are typically recorded in Target Lesion and Non-Target Lesion eCRFs at baseline. New Lesions 
appearing at post-baseline visits are recorded on an additional New Lesions eCRF. Figure 2 shows the 
mapping from a sample Target Lesion Details eCRF to the relevant TU and TR domains. These details 
will be common to all three lesion details in eCRFs. 

 

Figure 2. Mapping for TR and TU Domains  

TUMOR RESULTS (TR) 

The Tumor Results (TR) domain records quantitative measurements and qualitative assessments of 
tumors identified in the TU domain. These measurements are taken at baseline and each subsequent 
assessment to support response evaluations. Unlike the TU domain, the TR domain does not include 
anatomical location information. 

Key variables in the TR domain include: 

• TRLNKID, TRGRPID, TREVAL, TRMETHOD: Function similarly to their TU domain counterparts. 

• TRMETHOD and TUMETHOD: Usually consistent throughout the trial. 

• TRLNKGRP: Groups and links assessment records for response evaluation in the RS domain. 

• TRTEST/TRTESTCD: Store information about the test used for measurements. Table 1 presents 
commonly collected TESTCDs for solid tumor studies using RECIST. 

• TRORRES: Contains the original measurement result. 

• TRORRESU: Corresponding unit of measurement. 

• TRSTAT: Indicates if an assessment was not performed. 

• TRREASND: Provides the reason for an undone assessment. 

These variables ensure accurate tracking and evaluation of tumor measurements throughout the study. 
Figure 3 shows additional measurement variables collected on eCRF and mapped to TR domain.   

TRTESTCD TRTEST Description 

DIAMETER Diameter The length of a straight line passing through 
the center of a circle or sphere and connecting 
two points on the circumference. (NCI) 

LDIAM Longest Diameter The longest possible length of a straight line 
passing through the center of a circular or 
spheroid object that connects two points on the 
circumference. 

LPERP Longest Perpendicular / 
Short Axis Diameter 

 

The longest possible straight line or plane 
through a body or figure that is at a right angle 
to a given 

line or plane. 

SUMDIAM Sum of Diameter A calculation of the aggregated diameter 
values 

Table 1. Commonly collected TRTESTCD  

 

Figure 3. Mapping for TR Domains  

RESPONSE (RS) 

The Response (RS) dataset plays a crucial role in oncology clinical trials, particularly in studies evaluating 
solid tumors. It serves as a fundamental component in efficacy analyses by capturing tumor response 
assessments based on standardized criteria such as the Response Evaluation Criteria in Solid Tumors 
(RECIST). The RS dataset forms the basis for generating key analysis datasets used in efficacy 
evaluations, including ADaM datasets such as ADTR (Tumor Response), ADRS (Response) and ADTTE 
(Time-to-Event). 

The RS dataset is derived from case report form (CRF) data, which includes investigator-assessed and, 
in some cases, centrally reviewed tumor response evaluations. Each record in the RS dataset 
corresponds to a response assessment at a given time point, typically based on imaging scans. The 
dataset provides a standardized summary of tumor response at each evaluation time point, facilitating the 
derivation of key efficacy outcomes. 

Key variables in the RS dataset include: 

• VISITNUM (Visit Number) and VISIT (Visit Label) 

• RSDTC (Date of Response Assessment) 

• RSEVAL (Evaluator of Response, e.g., Investigator, Independent Review Committee) 

• RSCAT (Response Category, e.g., Target Lesions, Non-Target Lesions, Overall Response) 

• RSSTRESC (Standardized Response Result, e.g., CR, PR, SD, PD) 

• RSSTRESN (Numeric Representation of Response) 

• RSSTAT (Response Status, indicating whether the assessment was performed) 

• RSDY (Study Day of Response Assessment) 

The RS dataset is integral to oncology trials because it consolidates tumor burden changes from the TR 
dataset into discrete response categories. This enables the derivation of key efficacy endpoints such as 
progression-free survival (PFS) and overall response rate (ORR), which are critical in oncology studies. 
Figure 44 shows an example of a Disease Response CRF and the mapping to the SDTM RS domain.  

 

Figure 4. Mapping for RS Domain 

DESTINATION TLF: HOW DO WE GET THERE? 

In this section, we will present common efficacy summaries and figures with annotations from the ADaM 
datasets with discussion of the relevant ADaM derivations and associated SAS® code. 

BEST OVERALL RESPONSE 

Table 2 presents an example of a Summary of Best Overall Response. This provides a descriptive 
overview of subject response to treatment while on study. 

Disease Type 

Best Overall 

Response 

(ADRS.BOR) 

%  

(95% CI; n) 

Breast  

(N=X) 

Lung  

(N=X) 

Stomach   

(N=X) 

Pancreas 

(N=X) 

Liver 

(N=X) 

Total 

(N=X) 

CR xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

PR xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

SD xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

PD xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

NE xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

Table 2. Summary of Best Overall Response 

The ADaM dataset ADRS is programmed to determine best overall response (BOR) among all post-
baseline responses recorded for each subject and mapped to the RS domain in SDTM.  

Considerations:  

• To establish a best overall response (BOR), a subject must have an evaluable baseline assessment 
and at least one evaluable post-baseline assessment. Subjects not meeting these criteria are usually 
assigned the best overall response of NE.  

• Disease assessments after the start of subsequent anti-cancer therapy are typically excluded from 
determination of BOR.  

• In early phase trials, confirmation of partial or complete response at least 4 weeks after the first 
occurrence is recommended by RECIST and often required per protocol. This needs to be addressed 
in ADaM logic for determination of BOR. There is also a suggestion in RECIST that stable disease 
should last 6-8 weeks prior to being counted as BOR – this should be confirmed by programmer with 
protocol or sponsor.  

See Table 3 for an example from the value-level metadata tab of the ADaM specs for derivation of Best 
Overall Response. (Note: TMRESP refers to timepoint response, indicating the response determination at 
a given timepoint. SACDT is the subsequent anti-cancer therapy date.) 

PARAMCD PARAM Method  

BOR Best 
Overall 
Response 

For each subject: Sort by ADT. Exclude records where ADT > SACDT. Derive BOR based 
on TMRESP across all AVISIT for each subject: order unique TMRESP best to worst (CR, 
PR, SD, PD, NE, NA) and pick the best. If subject has no records with TMRESP or no 
records where ADT>SACDT then set BORA to NE.  

CBOR Confirmed 

Best 
Overall 
Response 

Only include records that occur prior to SACTDT: 

Use AVALC values where PARAMCD='TMRESP' in deriving this variable.  
(Check for BOR=CR): 
1. If TMRESP='CR' and subsequent assessment at least 28 days later (RS.RSDTC subsequent 
CR assessment - RS.RSDTC first CR assessment >=28) is also CR then set to 'CR'. Set ADT 
to date of first CR.  

Else 
(Check for BOR=PR): 
2. If TMRESP='PR' and subsequent assessment at least 28 days later (RS.RSDTC subsequent 
CR assessment – RS.RSDTC first CR assessment >=28) is in ('PR', 'CR') then set to 'PR'. Set 

ADT to date of first PR.  
Else 
(Check for BOR=SD): 
If at least one of the conditions in 3-7 are met, set to ‘SD’ and set ADT to min (ADT) in 3-7. 
3. If TMRESP='CR' and subsequent assessment less than 28 days later (RS.RSDTC 
subsequent assessment - RS.RSDTC first assessment<28) is 'CR' then set to 'SD'. Set ADT to 
date of first CR.  
4. If TMRESP='PR' and subsequent assessment less than 28 days later (RS.RSDTC 

subsequent assessment - RS.RSDTC first assessment<28) is in ('PR', 'CR') then set to 'SD'. 
Set ADT to date of first PR.  
5. If TMRESP='SD' and ADY>=42 then set to 'SD' and ADT to date of first SD.  
6.  If TMRESP='CR' and ADY>=42 and subsequent assessment in ('PR', 'SD', 'PD', 'NE', 
missing) then set to 'SD'. Set ADT to date of CR. 
7.  If TMRESP='PR' and ADY>=42 and subsequent assessment in (‘SD’, 'PD', 'NE', missing) 
then set to 'SD'. Set ADT to date of PR. 
Else  

8. If TMRESP='CR' and ADY<42 and subsequent assessment in (‘PR’, ‘SD’, ‘PD’) then set 
to 'PD" and ADT to date of subsequent assessment.  
Else 
(Check for BOR=PD): 
9.  IF TMRESP='PD' then set to 'PD' and ADT to date of first PD. 
Else set to “NE”  

Table 3. ADaM Specifications for Best Overall Response 

If the protocol requires confirmation of response, the derivation becomes more complicated, as shown in 
Table 3. If the initial response of CR or PR is not confirmed at a subsequent assessment at least 4 weeks 
later, then the confirmed best overall response is derived as SD. (Note: This example assumes that a 
subsequent assessment of at least 4 weeks (28 days) is required for confirmation of response and that 
stable disease of at least 6 weeks (42 days) is required for confirmed best overall response of SD.)  

Sample code is presented below for derivation of Confirmed Best Overall Response (Program 1). 

/* Identify CR and PR assessments and their confirmation */ 

data ADRS_CBOR; 

    set ADRS; 

    by USUBJID ADT; 

 

    /* Exclude records beyond SACDT */ 

    if ADT > SACDT then delete; 

 

    /* Assign response order */ 

    if TMRESP = "CR" then RESP_ORDER = 1; 

    else if TMRESP = "PR" then RESP_ORDER = 2; 

    else if TMRESP = "SD" then RESP_ORDER = 3; 

    else if TMRESP = "PD" then RESP_ORDER = 4; 

    else if TMRESP = "NE" then RESP_ORDER = 5; 

    else if TMRESP = "NA" then RESP_ORDER = 6; 

     

run; 

 

/* Create a dataset to check confirmation status */ 

proc sql; 

    create table CR_PR_Confirm as 

    select a.USUBJID, a.ADT as FIRST_ADT, b.ADT as CONFIRM_ADT,  

           a.TMRESP as FIRST_RESP, b.TMRESP as CONFIRM_RESP 

    from ADRS_CBOR a 

    left join ADRS_CBOR b 

    on a.USUBJID = b.USUBJID  

    and b.ADT >= a.ADT + 28 

    where a.TMRESP in ("CR", "PR"); 

quit; 

 

/* Derive CBOR */ 

data ADRS_CBOR_FINAL; 

    merge ADRS_CBOR CR_PR_Confirm; 

    by USUBJID; 

 

    length CBOR $2; 

 

    if TMRESP = "CR" and CONFIRM_RESP = "CR" then CBOR = "CR"; 

    else if TMRESP = "PR" and CONFIRM_RESP in ("PR", "CR") then CBOR = "PR"; 

    else if TMRESP = "CR" and (CONFIRM_ADT - FIRST_ADT) < 28 then CBOR = 

"SD"; 

    else if TMRESP = "PR" and (CONFIRM_ADT - FIRST_ADT) < 28 then CBOR = 

"SD"; 

    else if TMRESP = "SD" and ADY >= 42 then CBOR = "SD"; 

    else if TMRESP = "CR" and ADY >= 42 and CONFIRM_RESP in ("PR", "SD", 

"PD", "NE", "") then CBOR = "SD"; 

    else if TMRESP = "PR" and ADY >= 42 and CONFIRM_RESP in ("SD", "PD", 

"NE", "") then CBOR = "SD"; 

    else if TMRESP = "CR" and ADY < 42 and CONFIRM_RESP in ("PR", "SD", 

"PD") then CBOR = "PD"; 

    else if TMRESP = "PD" then CBOR = "PD"; 

    else CBOR = "NE"; 

run; 

Program 1. Derivation of Confirmed Best Overall Response 

Example code for calculation of 95% Confidence Interval using the binomial option in PROC FREQ as it 
applies to a Best Overall Response Table included below (Program ). 

/* Step 2: Calculate 95% Confidence Intervals */ 

ods output BinomialCLs=binomial_ci; 

proc freq data=bor_flag; 

    tables BOR / binomial(level='CR' exact) alpha=0.05; 

    by DISEASE_TYPE; 

run; 

 

/* Repeat for PR, SD, PD, and NE */ 

ods output BinomialCLs=binomial_pr; 

proc freq data=bor_flag; 

    tables BOR / binomial(level='PR' exact) alpha=0.05; 

    by DISEASE_TYPE; 

run; 

Program 2. SAS Code for Best Overall Response Table  

OBJECTIVE RESPONSE RATE  

Objective Response Rate as presented in Table 4 is defined as the number of subjects achieving a best 
overall response of CR or PR divided by the total number of subjects in the analysis population. Subjects 
with a best overall response of ‘NE’ are included in the denominator.  

 Disease Type 

 

Breast 

(N=X) 

Lung 

(N=X) 

Stomach 

(N=X) 

Pancreas 

(N=X) 

Liver 

(N=X) 

Total 

(N=X) 

Objective Response Rate (%) 

(95% CI; n) ADRS.ORR 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

xx.x 

(xx.x, xx.x; x) 

Table 4. Summary of Objective Response Rate 

A related endpoint is disease control rate, defined as the number of subjects achieving a best overall 
response of CR, PR, or SD. 

Table 5 shows an example of value-level metadata for derivation of Objective Response and Disease 
Control.  

PARAMCD PARAM Method  

COR Confirmed Objective Response 
Set to 'Y' if PARAMCD='CBOR' and AVALC in ('CR', 'PR').  
Else set to 'N' 

DCR Disease Control  
Set to 'Y' if PARAMCD='CBOR' and AVALC in ('CR', 'PR', 'SD').  
Else set to 'N' 

Table 5. ADaM Specifications for Objective Response and Disease Control 

TIME-TO-EVENT ENDPOINTS  

Time-to-event (TTE) endpoints in clinical research measure the time until a specific event, such as 
disease progression or death, occurs. The ADTTE dataset, structured according to CDISC ADaM 
standards, organizes this data with key variables like event dates, censoring indicators, and time-to-event 
values. Common TTE endpoints include progression-free survival (PFS), overall survival (OS), disease-
free survival (DFS), and duration of response (DOR). 

Guidelines for development of ADTTE are included in the CDISC document titled ‘The ADaM Basic Data 
Structure for Time-to-Event Analyses.’ ADTTE contains one record per parameter per subject. The 
variables used to characterize a time-to-event parameter for a given subject are provided below, along 
with their CDISC definitions:  

• STARTDT: The original date of risk for the time-to-event analysis. This is generally the time at which 
a subject is first at risk for the event of interest evaluation (as defined in the Protocol Disease 
assessments after start of subsequent anti-cancer therapy are typically excluded from determination 
of BOR.  

• AVAL: AVAL is the elapsed time to the event of interest from the origin. For example, if AVAL is 
measured in days, AVAL would be ADT – STARTDT or ADT – STARTDT + 1. 

• ADT: Analysis date/time of event or censoring associated with AVAL in numeric format. 

• CNSR: CNSR=0 for event and CNSR>0 for censored records. 

• EVNTDESC: Describe the event of interest or an event that warrants censoring 

• CNSDTDSC: Describe the circumstance represented by the censoring date if different from the event 
date that warrants censoring. 

Progression-Free Survival 

Progression-Free Survival (PFS) is defined as the duration of time from randomization (or first dose of 
study drug in non-randomized studies) to the first documentatoin of progressive disease or death, 
whichever occurs earlier. Subjects not experiencing either event are censored according to a set of 
censoring rules defined in the Statistical Analysis Plan (SAP). An example of a PFS summary table is 
shown below (Table 6).  



10 

Disease Type 

 

Breast 

(N=X) 

Lung 

(N=X) 

Stomach 

(N=X) 

Pancreas 

(N=X) 

Liver 

(N=X) 

Total 

(N=X) 

Number of Events (%) x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) 

ADTTE where PARAMCD=’PFS’ and 

CNSR=0 

      

Number of Censored (%)  

ADTTE where PARAMCD=’PFS’ and 

CNSR=1 

x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) 

Censoring Reason 

ADTTE.CNSDTDSC 

      

    No PD or Death Prior to Last  

    Evaluable Assessment 

x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) 

    No Post-Baseline Assessment x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) 

    Other Antitumor Treatment x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) x (xx.x%) 

       

Kaplan-Meier Estimates (days)        

    Median (95% CI) xx 

(xx,xx) 

xx 

(xx,xx) 

xx 

(xx,xx) 

xx 

(xx,xx) 

xx 

(xx,xx) 

xx 

(xx,xx) 

       

Survival Estimates, % (95% CI)       

   At 30 days  xx.x xx.x xx.x xx.x xx.x xx.x 

   At 60 days  xx.x xx.x xx.x xx.x xx.x xx.x 

   At 90 days  xx.x xx.x xx.x xx.x xx.x xx.x 

Table 6. Summary of Progression-Free Survival  

Table 7 shows an example of censoring rules for PFS. There are variations in censoring rules and studies 
will often specify the analysis using a primary set of censoring rules and an additional set of rules for 
sensitivity analysis. It is important to consult the protocol and SAP prior to writing specifications or code 
for censoring time-to-event endpoints. The FDA Guidance ‘Clinical Trial Endpoints for the Approval of 
Non-Small Cell Lung Cancer Drugs and Biologics’ is a helpful reference that provides multiple examples 
of PFS censoring schemes for both primary and supportive PFS analyses.  

Situation  Outcome Date of Event / Censoring  

New anticancer therapy initiated prior to disease 
progression or death Censor (non-event)  

Date of last adequate tumor assessment before new 
anticancer therapy is initiated  

Death before first planned assessment  Event  Date of Death 

No death before first planned assessment, and no 
evaluable post-baseline assessments  Censor (non-event) Date of randomization or first dose 

≥ 2 consecutive missed or non-evaluable tumor 
assessments prior to progression or death  Censor (non-event) 

Date of last adequate tumor assessment 
documenting no progression 

Death or progression after ≤ 1 missed or non-
evaluable tumor assessments  Event  Date of PD or Death  

No progression or death, no new anticancer therapy 

initiated  Censor (non-event)  Date of last adequate tumor assessment  

Table 7. Censoring Rules for PFS 

Example ADaM specification logic for PFS using censoring rules in Table 7 is provided below (Table 8). 

We use date of first dose as applicable in this example. In a randomized study, this would instead be date 

of randomization.  

Note:  
SACDT = Date of Subsequent Anti-Cancer Therapy 
DTHDT= Date of Death 
LEADT = Date of Last Evaluable Assessment 
PDDT = Date of first documented disease progression 
ADY=Analysis Study Day 
TRTSDT = Date of First Dose  
 
For the purposes of this example, we have assumed that the first planned assessment occurs at Week 6 
and assessments are expected every 6 weeks (42 days) while a subject remains on study.  
 

Condition ADT CNSR EVNTDESC CNSDTDSC 

1. If SACT^=. And SACTDT < min 
(PDDT DTHDT) 

LEADT 1 NEW ANTICANCER 
THERAPY 

LAST EVALUABLE 
ASSESSMENT PRIOR TO 
NEW THERAPY  

2. else if missing LEADT:  . < 
DTHDT  < 42  

        

               a. if . < DTHDY < 42 DTHDT  0 DEATH    

               b. if DTHDY >= 42 TRTSDT 0 FIRST DOSE NO EVALUABLE 
ASSESSMENTS AFTER 
BASELINE 

3. else if ADY > 126 and the there is 
not at least one non-NE assessment 
within 12 weeks prior (ADY - 84). 

max (last non-NE 
assessment prior to 
ADY, TRTSDT) 

1 >= 2 MISSED 
ASSESSMENTS 
PRIOR TO PD OR 
DEATH 

LAST EVALUABLE 
ASSESSMENT 
DOCUMENTING NO 
PROGRESSION 

4. else if not missing (PDDT) PDDT 0 PROGRESSIVE 
DISEASE  

  

5. else if not missing (DTHDT) DTHDT  0 DEATH    

6. else if not missing (LEADT) LEADT 1 NO PROGRESSIVE 
DISEASE OR DEATH 

LAST EVALUABLE 
ASSESSMENT 

DOCUMENTING NO 
PROGRESSION 

Table 8. ADaM Specifications for PFS 

See below for code that shows an example of code used to derive the PFS parameter (Program 3) 

/* Step 1: Define PFS Date (ADT) and Censoring Flag (CNSR) */ 

data pfs_data; 

    set adtte;  

 

    /* Initialize Variables */ 

    ADT = .; 

    CNSR = .; 

    EVNTDESC = ""; 

 

    /* Step 2: New Anti-Cancer Therapy Before PD or Death */ 

    if not missing(SACDT) and SACDT < min(PDDT, DTHDT, LEADT) then do; 

        ADT = SACDT; 

        CNSR = 1; 

        EVNTDESC = "OTHER ANTITUMOR TREATMENT"; 

    end; 
     

    /* Step 3: Check for PD or Death with Missed Assessments */ 

    else if not missing(PDDT) or not missing(DTHDT) then do; 

        ADY = min(PDDT, DTHDT) - TRTSDT + 1; 

 

        /* Check if subject had >=2 missed assessments before PD/Death */ 

        if acstage in ("Schedule 1(Q1/3W)", "Schedule 2(Q2/3W)") then do; 

            if (ADY < 211 and last_non_ne_date < ADY - 133) or 

               (211 <= ADY < 295 and last_non_ne_date < ADY - 154) or 

               (295 <= ADY < 379 and last_non_ne_date < ADY - 175) or 

               (379 <= ADY < 506 and last_non_ne_date < ADY - 217) or 

               (ADY >= 506 and last_non_ne_date < ADY - 259) then do; 

                CNSR = 1; 

                ADT = max(last_non_ne_date, TRTSDT); 

                EVNTDESC = "MISSED ASSESSMENTS PRIOR TO PD OR DEATH"; 

            end; 

        end; 

        else if acstage = "Schedule 3 (Q3/4W)" then do; 

            if (ADY < 197 and last_non_ne_date < ADY - 119) or 

               (197 <= ADY < 281 and last_non_ne_date < ADY - 147) or 

               (281 <= ADY < 364 and last_non_ne_date < ADY - 175) or 

               (364 <= ADY < 477 and last_non_ne_date < ADY - 203) or 

               (ADY >= 477 and last_non_ne_date < ADY - 231) then do; 

                CNSR = 1; 

                ADT = max(last_non_ne_date, TRTSDT); 

                EVNTDESC = "MISSED ASSESSMENTS PRIOR TO PD OR DEATH"; 

            end; 

        end; 

    end; 

 

    /* Step 4: Progression Event */ 

    else if not missing(PDDT) then do; 

        ADT = PDDT; 

        CNSR = 0; 

        EVNTDESC = "PROGRESSION"; 

    end; 

 

    /* Step 5: Death Event */ 

    else if not missing(DTHDT) and missing(SACDT) then do; 

        ADT = DTHDT; 

        CNSR = 0; 

        EVNTDESC = "DEATH"; 

    end; 

 

    /* Step 6: No PD, Last Evaluable Assessment */ 

    else if not missing(LEADT) then do; 

        ADT = LEADT; 
        CNSR = 1; 

        EVNTDESC = "NO PD"; 

    end; 

 

    /* Step 7: No Assessments After Treatment Start */ 

    else if missing(LEADT) and missing(SACDT) then do; 

        ADT = TRTSDT; 

        CNSR = 1; 

        EVNTDESC = "NO ASSESSMENT AFTER BASELINE"; 

    end; 

 

run; 

Program 3. SAS Code for PFS in ADTTE 

The Kaplan-Meier estimates of median survival and associated 95% CI, as well as the survival estimates 
(%) at the indicated timepoints displayed in the table above can be determined using proc lifetest in SAS.  

Example code is below (Program 4). The timelist option will give the survival estimates and 95% 
confidence intervals at the indicated timepoints. 

Duration of Response and Time to Response  

Duration of Response (DOR) and Time to Response (TTR) are typically derived only for subjects 
achieving an objective response (i.e. BOR or CBOR of CR or PR). 

TTR is defined as the time from randomization (or first dose of study drug if non-randomized study) to the 
first documentation of objective response.  

DOR is defined as the time from the date of first documentation of objective response to the date of 
disease progression or death from any cause, whichever is earlier.  

An example of a summary table for DOR and TTR is shown as Table 9 below. 

 Disease Type 

Endpoint 

Breast  

(N=X) 

Lung 

(N=X) 

Stomach  

(N=X) 

Pancreas  

(N=X) 

Liver 

(N=X) 

Total  

(N=X) 

Median Time to 
Response (TTR) in 

months  

xx.x xx.x xx.x xx.x xx.x xx.x 

    Mean (SD) xx.x (x.xx) xx.x (x.xx) xx.x (x.xx) xx.x (x.xx) xx.x (x.xx) xx.x (x.xx) 

    (Min, Max) (xx.xx, xx.xx) (xx.xx, xx.xx) (xx.xx, xx.xx) (xx.xx, xx.xx) (xx.xx, xx.xx) (xx.xx, xx.xx) 

ADTTE.AVAL 

where 
PARAMCD=’TTR’ 

      

       

Median Duration of 
Response (DOR) in 

months (Min, Max) 

ADTTE.AVAL 
where 
PARAMCD=’DOR’ 

xx.x 
(xx.x*, xx.x) 

xx.x 
(xx.x, xx.x) 

xx.x 
(xx.x, xx.x) 

xx.x 
(xx.x, xx.x) 

xx.x 
(xx.x, xx.x*) 

xx.x 
(xx.x, xx.x) 

        

Note: Kaplan Meier method (Lifetest procedure) has been used to calculate Median Duration of Response. 

Means procedure has been used to calculate Median Time to Response. 
* indicates the minimum or maximum represents a censored value.  

Table 9. Summary of DOR and TTR 

Censoring rules for DOR will be specified in the SAP and are usually similar to those described above for 
PFS. TTR can be summarized using descriptive statistics (mean, median, min, max) without 
consideration of censoring since all included subjects experience the event (response). Example code for 
generating the statistics needed to create Table 9 is displayed in Program 4 below. 

/* Step 1: Calculate Time to Response (TTR) Statistics */ 
proc means data=adtte n mean std min max maxdec=2; 

    where PARAMCD = 'TTR'; /* Time to Response */ 

    class DISEASE_TYPE; 

    var AVAL; /* Analysis Value */ 

    output out=ttr_stats  

        n=N 

        mean=Mean_TTR 

        std=SD_TTR 

        min=Min_TTR 

        max=Max_TTR; 

run; 

 

/* Step 2: Calculate Duration of Response (DOR) using Kaplan-Meier */ 

ods output Quartiles=km_dor; /* Store Kaplan-Meier quartiles */ 

proc lifetest data=adtte method=km plots=survival; 

    where PARAMCD = 'DOR'; /* Duration of Response */ 

    strata DISEASE_TYPE; 

    time AVAL*CNSR(1) / timelist(30 60 90 120 180); /* Specify time points 

for survival estimates */     

run; 

Program 4. Sample code for TTR and DOR table 

Additional time-to-event endpoints are defined below:  

• Time to Treatment Failure (TTF): time from randomization (or first dose of study drug if non-
randomized study) to treatment discontinuation for any reason, including progressive disease, 
toxicity, patient decision or death. 

• Disease-Free Survival (DFS): time from randomization (or first dose of study drug if non-randomized 
study) to disease recurrence, development of a new cancer, or death from any cause, whichever 
occurs earlier. 

• Overall Survival (OS): time from randomization (or first dose of study drug if non-randomized study) 
to death from any cause. 

Programming for these parameters will be similar to what is described above. Refer to SAP for study-
specific definitions and censoring rules. 

FIGURES 

Kaplan-Meier 

Kaplan-Meier figures are a visual representation of the Kaplan-Meier analysis described above. Time is 
represented on the x-axis, and the survival estimate on the y-axis. The line drops at the point that one or 
more events occur – subjects censored in between events are indicated with a ‘+’ symbol. See example 
in Figure 5.  

 

Figure 5. Kaplan-Meier Plot of PFS 

Example code to generate KM plot for PFS is shown below in Program 5. 

proc sort data=adam.adtte  out=adtte; 

by usubjid; 

where paramcd="PFS"; 

run; 

 

ods output Survivalplot=SurvivalPlotData quartiles = quart; 

proc lifetest data=adtte0(where=(trt=7)); 

   time aval*CNSR(1); 

   strata trt / test=logrank adjust=sidak; 

   run; 

 

proc sgplot data=SurvivalPlotData DATTRMAP=attrmap; 

symbolchar name=sym1 char='275A'x  /  textattrs=(Weight=Bold); 

  step x=time y=survival / group=stratum attrid=bar name='s'; 

  scatter x=time y=censored / markerattrs=(symbol=plus Color=Black size=10) 

MARKERCHARATTRS=( Weight=bold) name='c'; 

  scatter x=time y=censored / markerattrs=(symbol=plus Color=Black size=10) 

MARKERCHARATTRS=( Weight=bold) GROUP=stratum; 

  scatter x=time y=censor2 / markerattrs=(symbol=sym1 Color=Black size=18) 

name="d"; 

  xaxistable risk1 / x=timept class=stratum colorgroup=stratum 

valueattrs=(weight=bold); 

  xaxis type=linear Label="Time From First Dose (Months)" values=(0 to 12 by 

2) min=0 max=12  offsetmax=0.01 offsetmin=0.01 minor minorcount=1; 

yaxis type=linear  Label="Progression-free survival Estimate(%)" values=(0 

to 1 by 0.1) ;   

  keylegend 'c' 'd'/ title = "Censored: " across = 1 location=inside 

position=topright titleattrs=(Weight=Bold) noborder; 

   inset "MEDIAN (95% CI): &MED."/border position=topright TEXTattrs=( 

size=10pt); 

  format Stratum trtf.; 

run; 

Program 5: Sample code to generate KM-plot for PFS.  



16 

Waterfall Plot  

The waterfall plot displays the best (i.e. smallest) percent change from baseline in the sum of target lesion 
diameters for each subject. The plot is typically arranged from highest to lowest best percent change and 
can be annotated or color-coded with best overall response and/or disease type. See example in Figure 
66. 

  

Figure 6. Waterfall Plot 

The sum of diameters and best percent change from baseline needed to generate this figure are derived 
in ADTR. The sum of diameters at each post-baseline visit should only be derived in ADTR if all of the 
target lesions measured at baseline were measured at that visit. See Table 10 below for an example of 
ADaM specifications for the relevant fields in ADTR.  

Variable  Label  Type Method  

PARAM Parameter text Sum of Diameters 

PARAMCD Parameter Code text SUMDIAM 

AVAL Analysis Value float  

Only include records that occur prior to 
SACTDT: 
Equals sum of AVAL where PARAMCD in 
T1DIAM-T5DIAM. For visits after 
baseline, only calculate if the same 
number of target lesion have 
measurements recorded as at baseline. If 
there are less, then set to 'NE'. 

ADT Analysis Date  integer Set to TR.TRDTC.   

BASE Baseline Value float  
Set to AVAL where 
PARAMCD='SUMDIAM' and ABLFL='Y' 

CHG Change from Baseline float  
Where PARAMCD='SUMDIAM', 
CHG=AVAL-BASE 

PCHG  Percent Change from Baseline float  
Where PARAMCD='SUMDIAM', 
PCHG=(CHG/BASE)*100 

ABLFL Baseline Record Value  text  

Where PARAMCD='SUMDIAM', set to 'Y' 
on the record with the latest non-missing 
AVAL where ADT<=TRTSDT.  

ANL01FL Analysis Flag 01  text 

Set to 'Y' for the smallest value of PCHG 
where PARAMCD='SUMDIAM'. Else set 
to null.  

Table 10. Specifications for Sum of Diameters and Best Percent Change in ADTR 

Example code to generate waterfall plot is shown below in Program 6. 

data final; 

    merge adam.adtr(where=(paramcd='SUMDIAM' & anl01fl='Y') in=a) 

          adsl(in=b) 

          adam.adrs(where=(paramcd='BOR' & anl01fl='Y') rename = 

avalc=bestresp); 

    by usubjid; 

    if a & b; 

    if missing(aeotdt) then ongoing=atrtdurd+1; 

    length subj_var org_var $50. ; 

 

    subj_var="'"||strip(subjid)||"'"; 

    org_var="'"||strip(tumorg)||"'"; 

    if pchg=>0 then marker=pchg+4; 

    else if pchg<0 then marker=pchg-5; 

run; 

proc sort data=final; 

    by descending pchg subjid; 

run; 

 

proc template; 

    define statgraph seriesplot; 

    begingraph; 

 

    discreteattrmap name='symbols' / ignorecase=false trimleading=true                

discretelegendentrypolicy=attrmap; 

    value "CR" / fillattrs=(color=blue) markeratttrs=(color=black); 

    value "PR" / fillattrs=(color=green) markeratttrs=(color=black); 

    value "SD" / fillattrs=(color=orange) markeratttrs=(color=black); 

    value "PD" / fillattrs=(color=grey) markeratttrs=(color=black); 

 

    enddiscreteattrmap; 

    discreteattrvar attrvar=groupmarkers var=bestresp attrmap='symbols'; 

 

 

    layout gridded; 

        layout overlay / cycleattrs=false 

                         yaxisopts=(griddisplay=off 

                                    label='Best % Change in Sum of 

Diameters' 

                                    display=(line label ticks tickvalues) 



18 

                                    type=linear 

                                    linearopts=(viewmin=-100 viewmax=100 

tickvaluesequence=(start=-100 end=100 increment=10) tickvaluepriority=true)) 

                         xaxisopts=(griddisplay=off 

                                    label='' 

                                    display=(line ticks tickvalues) 

                                    type=discrete 

                                    

discreteopts=(tickvaluelist=(&subj_list.) tickdisplaylist=(&org_list.) 

tickvaluerotation=vertical)) 

        ; 

        scatterplot x=subjid y=marker / group=groupmarkers name='test2' 

markercharacter=groupmarkers; 

        barchart x=subjid y=pchg / group=groupmarkers name='test1'; 

        discretelegend 'test1' / valign=bottom halign=left location=inside 

across=1 displayclipped=true; 

        referenceline y=-30 / lineattrs=(pattern=2); 

 

        endlayout; 

    endlayout; 

 

    endgraph; 

    end; 

run; 

 

ods graphics / reset width=10.0in height=5.75in noborder noscale; 

ods rtf file="&titlekey..rtf" style=sasmonospace nogtitle nogfootnote; 

 

proc sgrender data=final2 template=seriesplot; 

run; 

ods _all_ close;  

ods graphics off; 

Program 6. Sample code to generate waterfall plot 

Spider Plot  

The spider plot displays the percent change from baseline in sum of target lesions by subject at each 
assessment. Each line represents the values measured for a single subject over time. Lines can be 
grouped by color for disease type, starting dose level, or any other subgroup of interest. See example in 
Figure 77. 

 

Figure 7. Spider Plot 

Example code to generate the spider plot is shown in Program 7. 

data adtr; 

    length group $50.; 

    set adam.adtr; 

    where paramcd='SUMDIAM' & refl='Y'  ; 

    if ablfl='Y' then ady=0; 

 

    group = subjid||" "||strip(put(DOSE01P, best.))||" "||strip(dose01u); 

run; 

 

proc transpose data=adtr out=final; 

    id tumorg; 

    var pchg; 

    by cohort usubjid ady; 

run; 

proc template; 

    define statgraph seriesplot; 

    begingraph; 

 

        legenditem type=markerline name='test1' / markerattrs=(color=red 

symbol=circle) lineattrs=(color=red pattern=solid) label='Breast'; 

        legenditem type=markerline name='test2' / markerattrs=(color=blue 

symbol=square ) lineattrs=(color=blue pattern=solid) label='Liver'; 

        legenditem type=markerline name='test3' / markerattrs=(color=green 

symbol=triangle) lineattrs=(color=green pattern=solid) label='Lung'; 

        legenditem type=markerline name='test4' / markerattrs=(color=purple 

symbol=diamond ) lineattrs=(color=purple pattern=solid) label='Ovary'; 

        legenditem type=markerline name='test5' / markerattrs=(color=orange 

symbol=star ) lineattrs=(color=orange pattern=solid) label='Other: 

Endometrium'; 

 

  layout gridded; 

         layout overlay / cycleattrs=false 

                         yaxisopts=(griddisplay=off 

                                    label='Change from Baseline (%)' 

                                    type=linear 

                                    linearopts=(tickvaluesequence=(start=-60 

end=60 increment=5) tickvaluepriority=true)) 

                         xaxisopts=(griddisplay=off 

                                    label='Time from Baseline (days)' 

                                    display=(line label ticks tickvalues) 

                                    linearopts=(tickvaluesequence=(start=0 

end=400 increment=50) tickvaluepriority=true)) 

        ; 

        referenceline y=-30 / lineattrs=(pattern=2); 

        referenceline y=0 / lineattrs=(pattern=1); 

 

 

        seriesplot x=ady y=breast / group=usubjid  

                                    groupdisplay=cluster  

                                    clusterwidth=0  

                                    display=all  

                                    name='marker2' 

                                    markerattrs=(color=red symbol=circle 

size=6pt) lineattrs=(color=red pattern=solid) 

  ; 

        seriesplot x=ady y=liver / group=usubjid  

                                    groupdisplay=cluster  

                                    clusterwidth=0  

                                    display=all  

                                    name='marker2' 

                                    markerattrs=(color=blue symbol=square 

size=6pt) lineattrs=(color=blue pattern=solid) 

  ; 
 

        seriesplot x=ady y=lung / group=usubjid  

                                    groupdisplay=cluster  

                                    clusterwidth=0  

                                    display=all  

                                    name='marker2' 

                                    markerattrs=(color=green symbol=triangle 

size=6pt) lineattrs=(color=green pattern=solid) 

  ; 

 

        seriesplot x=ady y=ovary / group=usubjid  

                                    groupdisplay=cluster  

                                    clusterwidth=0  

                                    display=all  

                                    name='marker2' 

                                    markerattrs=(color=purple symbol=diamond 

size=6pt) lineattrs=(color=purple pattern=solid) 

  ; 

 

        seriesplot x=ady y=OTHER__ENDOMETRIUM / group=usubjid  

                                    groupdisplay=cluster  

                                    clusterwidth=0  

                                    display=all  

                                    name='marker2' 

                                    markerattrs=(color=orange symbol=star 

size=6pt) lineattrs=(color=orange pattern=solid) 

  ; 

 

 

         endlayout; 

      endlayout; 

 

        layout globallegend / type=column; 

            discretelegend 'test1' 'test2' 'test3' 'test4' 'test5'/ 

valign=bottom; 

        endlayout; 

    endgraph; 

    end; 

run; 

 

ods rtf file="&titlekey..rtf" style=sasmonospace nogtitle nogfootnote; 

 

 proc sgrender data=final template=seriesplot; 

 by sched; 

run; 

 

ods _all_ close; 

ods graphics off; 

Program 7. Sample code to generate spider plot. 

Swimmer Plot  

The swimmer plot displays the duration of each subject on study drug with their confirmed best overall 
response markers at the time of tumor assessment during the course of study. Additional milestones such 
as death or adverse events leading to treatment discontinuation can be added. See example in Figure 
88. 

 

Figure 8. Swimmer Plot 

Example code to generate the swimmer plot is shown in Program 8. 

data final; 

    set adam.adrs; 

    where paramcd='CBOR' & ^missing(trt01pn); 

run; 

 

 

proc template; 

    define statgraph seriesplot; 

    dynamic _MAX_SUB _BAR_SIZE; 

    begingraph; 

    discreteattrmap name='symbols' / ignorecase=false trimleading=true 

discretelegendentrypolicy=attrmap; 

    value "CR" / markerattrs=(color=red symbol=circlefilled) ; 

    value "PR" / markerattrs=(color=blue symbol=squarefilled ) ; 

    value "SD" / markerattrs=(color=brown symbol=trianglefilled ) ; 

    value "PD" / markerattrs=(color=purple symbol=diamondfilled ) ; 

    value "NE" / markerattrs=(color=orange symbol=starfilled ) ; 

    enddiscreteattrmap; 

    discreteattrvar attrvar=groupmarkers var=avalc attrmap='symbols'; 

layout gridded; 

        layout overlay / cycleattrs=false 

                         yaxisopts=(griddisplay=off 

                                    label='Subjects who received Study Drug' 

                                    display=(line label) 

                                    type=linear 

                                    linearopts=(viewmin=0 viewmax=_MAX_SUB 

tickvaluesequence=(start=1 end=_MAX_SUB increment=1) 

tickvaluepriority=true)) 

                         xaxisopts=(griddisplay=off 

                                    label='Time Since First Dose (Days)' 

                                    display=(line label ticks tickvalues) 

                                    linearopts=(tickvaluesequence=(start=0 

end=500 increment=25) tickvaluepriority=true)) 

        ; 

vectorplot x=atrtdurd y=yvar xorigin=0 yorigin=yvar / group=trt_group1 

arrowheads=false includemissinggroup=false  datatransparency=0.3 

name='vector1'; 

 

vectorplot x=atrtdurd y=yvar xorigin=0 yorigin=yvar / group=trt_group2 

arrowheads=false includemissinggroup=false  datatransparency=0.3 

name='vector2'; 

 

vectorplot x=atrtdurd y=yvar xorigin=0 yorigin=yvar / group=trt_group3 

arrowheads=false includemissinggroup=false  datatransparency=0.3 

name='vector3'; 

 

vectorplot x=atrtdurd y=yvar xorigin=0 yorigin=yvar / group=trt_group4 

arrowheads=false includemissinggroup=false  datatransparency=0.3 

name='vector4'; 

 

scatterplot x=ady y=yvar / group=groupmarkers  name='scatter' 

includemissinggroup=false ; 

 

scatterplot x=eval(atrtdurd-atrtdurd) y=yvar / datalabel=subjid 

datalabelattrs=(color=black) markerattrs=(size=0) datalabelposition=left; 

 

scatterplot x=ongoing y=yvar / name='ong' legendlabel='Treatment Ongoing' 

markerattrs=(symbol=trianglerightfilled size=9 color=black); 

 

scatterplot x=dthdy y=yvar / name='dth' legendlabel='Death' 

markerattrs=(symbol=X size=9 color=black); 

 

endlayout; 

 

endgraph; 

end; 

run; 

 

 

ods rtf file="&titlekey..rtf" style=sasmonospace nogtitle nogfootnote; 

 

 proc sgrender data=final2 template=seriesplot; 

 dynamic _MAX_SUB=&last1. _BAR_SIZE=7; 

format AVALC $resp.; 

run; 

 

ods _all_ close; 

ods graphics off; 

Program 8. Sample code to generate swimmer plot. 

CONCLUSION  

We have arrived at our destination. We hope that this paper will provide a guide for new programmers 
and statisticians starting their journey in oncology trials. We have presented the foundational components 
of this vast landscape – we hope this will be a starting point to build confidence in tackling these initial 
milestones and even the increasing challenges of more complex designs and later phase trials - there is 
so far to go from here. Bon voyage! 

